

Zbatime kalkulusi në Maxima

Qëllimet dhe objektivat

- Shembuj funksionesh kalkulusi në Maxima
- Zbatime kalkulusi

Përmbajtja

- 1 Integrimi
- 2 Derivimi
- 3 Limiti
- 4 Seritë
- 5 Shembuj zbatimesh në kalkulus

Funksioni integrate

- Funksioni `integrate` mund të zbatohet për zgjidhjen analitike të një integrali të pacaktuar ose zgjidhjen numerike të një integrali të caktuar.
- Menyja *Calculus* > *Integrate...*
- Integrali i pacaktuar:

```
integrate(1/(1+x^2),x);  
'integrate(1/(1+x^2),x);  
ev(%,nouns);
```

Funksioni integrate (Vazhdim)

- Integrali i caktuar:

```
integrate(1/(1+x^2), x, alpha, beta);  
'integrate(1/(1+x^2), x, alpha, beta);  
ev(%,nouns);  
integrate(1/(1+x^2), x, 0, 2);  
integrate(1/(1+x^2), x, minf, inf);  
'integrate(1/(1+x^2), x, minf, inf);  
ev(%,nouns);  
integrate(exp(-x^2), x);  
integrate(exp(-x^2), x, 0, 1);  
float(%)  
integrate(exp(-x^2/2), x, 0, inf);  
float(%)  
integrate(exp(-x^2)*sin(x/2), x, 0, %pi/2);  
float(%)  
rectform(%)
```

Funksioni changevar

- Zëvendësimi i ndryshores:

```
'integrate(x^2/(1+x^2), x, 0, 2);  
changevar(%, x-tan(t), t, x);  
trigsimp(%);  
'integrate(1/sqrt(1+x), x);  
changevar(%, u-x-1, u, x);  
'integrate(1/sqrt(1-x^2), x);  
changevar(%, x-sin(t), t, x);
```

```
integrate(x^2/(1+x^2), x, 0, 2);  
integrate(1/sqrt(1+x), x);  
integrate(1/sqrt(1-x^2), x);
```

Funksioni differentiate

- Funksioni differentiate mund të zbatohet për gjetjen analitike të derivateve të rendeve të larta.
- Menyja *Calculus > Differentiate...*
- Derivati i rendit të parë:

```
diff(x^3*sin(x), x);
```

```
diff(x^3*sin(x), x, 3);
```

```
'diff(x^3*sin(x), x);
```

```
ev(%, nouns);
```

```
'diff(x^3*sin(x), x, 3);
```

```
ev(%, nouns);
```

Funksioni limit

- Funksioni limit mund të zbatohet për gjetjen e limitit të një funksioni.
- Menyja *Calculus > Find Limit...*
- Kalkulimi i një limiti:

```
limit(1/(x^2+1), x, 1);  
'limit(1/(x^2+1), x, 1);  
ev(%, nouns);  
limit((x^2-2*x+1)/(x-1), x, 1);  
'limit('diff(x^2-2*x+1,x)'/diff(x-1,x), x, 1);  
ev(%, nouns);
```

- Limitet e njëanshme:

```
limit(x+1/(x-2), x, 2);  
limit(x+1/(x-2), x, 2, minus);  
limit(x+1/(x-2), x, 2, plus);
```


Funksioni taylor

- Funksioni taylor mund të zbatohet për gjetjen e zbërthimit të Taylor-it të një funksioni.
- Menyja *Calculus > Get Series...*
- Zbërthimi i Taylor-it:

```
taylor(sin(x)+cos(x), x, 0, 8);
```

```
taylor(log(x), x, 1, 8);
```

```
taylor(sin(x), x, 0, 7);
```

Funksioni sum

- Funksioni `sum` mund të zbatohet për kalkulimin e shumave të fundme ose të serive.
- Menyja *Calculus > Calculate Sum...*
- Llogaritja e një shume të fundme:

```
sum(1/k^2, k, 1, 100);  
float(%)
```

- Llogaritja e shumës së një serie:

```
sum(1/k^2, k, 1, inf);  
sum(1/k^2, k, 1, inf), simpsum;  
float(%)
```

Funksioni product

- Funksioni product mund të zbatohet për kalkulimin e një prodhimi.
- Menyja *Calculus > Calculate Product...*
- Llogaritja e një prodhimi:

```
product(1/k, k, 1, n);
```

```
product(1/k, k, 1, 10);
```

```
product(k, k, 1, 5);
```

```
5!;
```

Shembuj zbatimesh në kalkulus

Shembull $(\lim_{x \rightarrow 0} \frac{\sin x}{x})$

Gjeni $\lim_{x \rightarrow 0} \frac{\sin x}{x}$

```
'limit(sin(x)/x, x, 0); ev(%, nouns);  
'limit(sin(x)/x, x, 0, minus); ev(%, nouns);  
'limit(sin(x)/x, x, 0, plus); ev(%, nouns);  
'limit('diff(sin(x),x)/'diff(x,x), x, 0);  
    ev(%, nouns);  
wxplot2d(sin(x)/x, [x,-10,10],  
    [xlabel,"x"], [ylabel,"y"]];
```

Shembuj

Shembull (Derivati si limit)

Me përkufizim, derivati i një funksioni $f(x)$ është limiti

$$\frac{df}{dx} = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}.$$

Llogaritni si limit derivatet vijuese $\frac{d}{dx} \sin x$, $\frac{d}{dx} \sqrt{x}$ dhe $\frac{d}{dx} \frac{1}{\sqrt{x}}$.

```
'limit((sin(x+h)-sin(x))/h, h, 0); ev(%, nouns);  
diff(sin(x),x);
```

```
'limit((sqrt(x+h)-sqrt(x))/h, h, 0); ev(%, nouns);  
diff(sqrt(x), x);
```

```
'limit((1/sqrt(x+h)-1/sqrt(x))/h, h, 0); ev(%, nouns);  
diff(1/sqrt(x),x);
```

Shembuj (Vazhdim)

Shembull (Derivimi implicit)

```
diff(1/sqrt(x),x);  
eq : x*u(x)+u(x)^2=sin(x);  
eqd : diff(eq, x);  
solve(eqd, diff(u(x),x));
```

Shembuj (Vazhdim)

Shembull (Gjetja e ekstremumeve dhe pikave të infleksionit)

```
f(x):=x^4-5*x^2+6*x+3;
eqmm : diff(f(x),x)=0;
solmm : solve(eqmm, x);
xc : makelist(rhs(s), s, solmm);
yc : f(xc);
wxplot2d([f(x), [discrete,xc,yc]], [x,-2,2],
  [style,[lines,2,1],[points,3,2,3]],
  [xlabel,"x"], [ylabel,"y"],
  [legend,"f(x)","p.krit."]);
wxplot2d([f(x), [discrete,xc,yc]], [x,0.7,1.1],
  [style,[lines,2,1],[points,3,2,3]],
  [xlabel,"x"], [ylabel,"y"],
  [legend,"f(x)","p.krit."]);
```

Shembuj (Vazhdim)

Shembull (Gjetja e ekstremumeve dhe pikave të infleksionit...)

```
f2 : diff(f(x),x,2);  
makelist(subst([x=crit], f2), crit, xc);  
float(%);  
solinfl : solve(f2=0, x);  
xi : makelist(rhs(s), s, solinfl);  
yi : f(xi);  
wxplot2d([f(x), [discrete,xc,yc], [discrete,xi,yi]],  
          [x,-2,2],  
          [style,[lines,2,1],[points,3,2,3],[points,3,2,1]],  
          [xlabel,"x"], [ylabel,"y"],  
          [legend,"f(x)","p.krit.","p.infl."]);
```


Shembuj (Vazhdim)

Shembull (Gjetja e ekstremumeve dhe pikave të infleksionit...)

```
wxplot2d([f(x), [discrete,xc,yc], [discrete,xi,yi]],  
          [x,0.85,0.95],  
          [style,[lines,2,1],[points,3,2,3],[points,3,2,1]],  
          [xlabel,"x"], [ylabel,"y"],  
          [legend,"f(x)","p.krit.","p.infl."]);  
wxplot2d([f(x), [discrete,xc,yc], [discrete,xi,yi]],  
          [x,-0.9129,-0.9128],  
          [style,[lines,2,1],[points,3,2,3],[points,3,2,1]],  
          [xlabel,"x"], [ylabel,"y"],  
          [legend,"f(x)","p.krit.","p.infl."]);  
f3 : diff(f(x),x,3);  
makelist(subst([x=infl], f3), infl, xc);  
float(%);
```

Shembuj (Vazhdim)

Shembull (Lëvizja dydimensionale me nxitim konstant)

```
eqdy : 'diff(y, t, 2) = a;  
eqdx : 'diff(x, t) = v0*cos(theta);  
soly : ode2(eqdy, y, t);  
eqy : ic2(soly, t=0, y=y0, 'diff(y,t)=v0*sin(theta));  
solx : ode2(eqdx, x, t);  
eqx : ic1(solx, t=0, x=x0);
```

Shembuj (Vazhdim)

Shembull (Derivatet dhe integralet)

```
'diff('integrate(exp(-x^2), x, 0, t),t);  
ev(%, nouns);  
'integrate('diff(exp(-x)*sin(x),x,2), x, 0, t);  
ev(%,nouns);  
integrate(%, t);  
ratsimp(%);  
expand(%);
```

Shembuj (Vazhdim)

Shembull (Formula derivimesh dhe integrimesh)

```
'diff(u(x)*v(x), x); ev(%, nouns);  
'diff(u(x)/v(x), x); ev(%, nouns);  
'integrate(x^n, x); ev(%, nouns);  
'integrate(x^-1, x); ev(%, nouns);
```

Shembuj (Vazhdim)

Shembull (Integralet e shumëfishta)

```
'integrate('integrate(x^2+y^2, y, 0, x), x, 0, 1);  
    ev(%, nouns);  
'integrate('integrate(exp(x+y), y, 0, 2), x, 0, 2);  
    ev(%, nouns);  
'integrate('integrate(x*sin(y), y), x);  
    ev(%, nouns);
```

Përfundim

- Integralet e caktuar dhe pacaktuar me anë të funksionit `integrate`
- Derivimi me anë të funksionit `diff`
- Kompjutimi i limiteve me anë të funksionit `limit`
- Zbatime në kalkulus